
El presentetutorial seorientaa introducira Holtek a los ingenierosy developersfamiliarizadoscon MCS-51, deuna
formaprácticay concisa.Paraun análisisde laarquitecturay descripción delusode los periféricosmás comunes, se
sugiere el tutorial CTU-005.

La arquitecturade Holtek es en ciertomodosimilar a la de MCS-51.Haciendoalgunassalvedades,podemosasumir
queHoltek esunaespeciedesubsetde laarquitecturaMCS-51, lo cual haremosparaesteanálisis,resaltandoaquellos
factores arquitectónicos que los acercan o diferencian:

En ambos:

�
El tradicionalacumulador A seve comoun SFRmás (ACC),permitiendoqueunagrancantidadde operaciones
sólo realizables sobre la memoria puedan efectuarse en el acumulador�
La arquitecturapermiteinterrupcionesvectorizadas, aunque enrealidadsetratade offsetsfijos. Una interrupción
ocasionaun salto aunaposiciónfija en memoria de instrucciones, dependiente de la causa de lainterrupción.El
contenido del PC seguarda en el stack(hardwareen Holtek, RAM interna en MCS-51)para permitir su
recuperación al ejecutaruna instrucciónde retorno,querestableceel estadode lasinterrupciones.No se salva el
contextode laCPU,el programadebesalvarlos registros necesarios enalgunaposiciónde memoria (o el stack en

MCS-51). Esposible prescindir de salvar el registro deestado(STATUS en Holtek, PSW en MCS-51) si el
interrupt handler realiza funciones clásicas.�
MCS-51 dispone de 4bancosde 8 registros enRAM interna,seleccionablesmedianteun SFR. El espaciode
memoria interna de 0x00 a 0x80 esdireccionabletanto directa como indirectamente(mediantedos de estos
registros,R0 y R1), mientrasque elespaciode 0x80 a 0xFF essolo direccionablede formaindirecta; los SFR
ocupan elespaciode 0x80 a 0xFF en direccionamiento directo.Holtek dispone deunaserie deSFRsy luegoRAM
interna, desde0x00 hasta0xFF, direccionableen forma directao indirecta medianteun par de SFRs(uno es
punteroy el accesoal otro realizala indirección,dado que no hayopcodesparadireccionamientoindirecto).Las
operaciones lógicas se realizan de forma indistinta sobre cualquier posición de memoria interna.�
Es posiblealojar tablaso constantes en memoria deprograma,accediéndolasmedianteuna instrucciónespecial
(MOVC en MCS-51,TABRDC/TABRDL enHoltek, mediantedireccionamiento indexado (PCo DPTR)en MCS-
51 e indirecto (TBLP) en Holtek.�
Ante una resta,el flag de Carry seutiliza como Borrow. MCS-51 presentaun borrow complementado,como
muchos micros: SUBB A , m � A � A � C � m ; Holtek presenta un borrow normal, como muchos DSP: 

SBC A , m � A � A � m � C � A � m � C .

Sólo Holtek:

�
Empleapipelining, todas las instruccionesdemoranun ciclo de máquina(cuatrociclos de clock) en ejecutarse,
excepto aquellas que obligan a purgar el pipeline, como instrucciones de salto.�
Utiliza técnicas deVLIW al incorporarlos datos en lainstrucción,es decir, la memoria de instrucciones es deun
ancho mayor, demodode poderincluir constantes(valoresinmediatos,direccionesde datos) en las instrucciones.
Por ejemplo, muchos micros de esta serie tienen un ancho de palabra de instrucciones de 14-bits. En este espacio se
distribuyen las más de 60 instrucciones disponibles,algunas de las cuales incluyenlos 8 bits del dato
correspondiente (direccionamientoinmediato)o su dirección en memoria de datos (direccionamiento directo). Esto
limita la cantidad contigua de memoria de datos disponible a 256 bytes� 28 � .�
No permitecargarun datoinmediatoen memoria sino que es necesariocargarloprimeroen el registro Ay luego
cargar éste en memoria.�
Tiene elprogramcounteraccesibleen la memoria de datos, esun SFRmás. Dado que la memoria de datos es de 8-
bits, los bits (14, 15ó 16) del PC están partidos enPCL (8-bits) y PCH (6, 7 ú 8 bits). El registroPCH no es
accesibledirectamente,por lo que las instrucciones que modifiquen elflujo mediantecálculo y operaciónsobre
PCL (por oposiciónal uso de CALL o JMP) estánlimitadas a un segmento de 256words de programa.Las
operacionesde CALL y JMP operansobrela totalidad de la memoria deprogramasin restriccionesni bank
switching.�
No posee instrucciones de manejo de stack.�
Afectael STATUSanteoperacionesaritméticasy lógicas solamente,moverun dato al acumulador noalterael flag
de Zero.�
No poseedireccionamiento directo de bity álgebrabooleanaen el sentido del MCS-51, sí permiteaccedera un bit
dentro de un byte.�
No dispone de memoria externa�
Dispone del acumuladorcomo operandoobligado en toda operaciónaritmética o lógica. El resultadopuede
depositarse en la memoria si ésta es el otro operando. Estoatenúael inconveniente de disponer desóloun registro
en la CPU, dado quepuedehacerse queuna determinadaoperación no altere elacumulador,guardandoel
resultado en la memoria de datos.

� 	 
 � � 
 � � � � �

MCS-51

Presenta direccionamientolineal hasta64K flash(programay tablas,accesoindexado)y 64K RAM (datos,acceso
indirecto). Existe un espacio de 256 bytes interno con acceso directo/indirecto o indirecto, según la dirección.

Se trata de un CISC conarquitecturaHarvard modificada, stack enRAM interna.DiferentesCPUsde diversos
fabricantes aceptanclocks de hastamás de 40 MHz,internamentedivididos por doce o no, lo que permiteuna
ejecuciónde más de 40MIPS pico, que suelendisminuir bastante en operación sostenida, dependiendo delalgoritmo,
dado que en los micros 'single-cycle' por lo general la mayoría de las operaciones se ejecutan en varios ciclos de clock.

Comoperiféricos,encontramosunavasta variedad dedispositivos,dependiendo de la inventiva del fabricante. Por lo
general todos disponen de una UART o dos y dos o tres timers de 8 ó 16-bits. 

Holtek series “Cost Effective”, “I/O” y “A/D”

  Presenta direccionamiento lineal hasta 8K OTP/MTP y 224 bytes RAM; el modelo que incorpora una cantidad mayor
de RAM emplea bank switching.

Setratadeun RISCconarquitecturaHarvard modificada, stack enhardwarede 2 a 16 niveles, según elmodelo.Su
operación es bastantesimilar a PICmicro de Microchip, pero sin eluso de bank switching en flashni I/O y
practicamentenulo en RAM. La CPU aceptaun clock de hasta8 MHz, internamentedividido por cuatro, lo que
permiteunaejecucióna 2MIPS pico, que no disminuyen demasiado en operación sostenida, dado que lagranmayoría
de las operaciones se ejecutan en un ciclo de clock.

Comoperiféricos,encontramosun timer de 8-bits en casitodoslos modelos;y ninguno,uno o dostimersde 16-bits
en las series “I/O”y “A/D”. Los timerstienencapacidad decontrolarun buzzermediantedossalidascomplementarias
con 50% deciclo de trabajo.Los modelosde la serie “A/D” incorporanun conversorAD de 9 ó 10-bits de 4 ú 8
canales,y los modelosmás avanzadosagreganinterfaz I2C y generadorde PWM. El watchdogtimer poseeun
oscilador independiente.

Paralelo de hardware

Dada lagranvariedad de micros MCS-51, no haremosunacomparativa decaracterísticassimilaresdehardwareen
estaoportunidad.Además, la mayoría delos MCS-51 del mercadoapuntana unagamamásalta. Es la intenciónde
este tutorial facilitar al migración en los casos en que un micro de gama más baja como Holtek sea una alternativa más
interesante.

Paralelo de software

  A continuación haremos un paralelo de software.
Las comparaciones deusode memoria deprogramaseránuno a uno. Si bien MCS-51 empleabytesy Holtek words,

comparamosambosprocesadorespor su capacidad en unidades elementales (1Kbyte vs. 1K word).Lo hacemos así
dado que la capacidad de memoria deprogramadeambosestáespecificadade esta forma,y nos permiterealizaruna
comparacióndirecta.La cantidaddeciclosestá basada enlos ciclosdemáquinadeun MCS-51standard,en el cualun
ciclo demáquinasondoceperíodosdeclock. Diferentes implementacionestendrándiferentetiming, y por lo general
los microssingle-cyclenecesitan más deun ciclo declock en muchas de lasoperaciones,por lo queunacomparación
contra un determinadomicro deberárealizarsetomando en cuenta sutiming particular.Respectoal assembler de
Holtek, el mismodeterminamediantelas declaraciones qué nombres corresponden a variablesy cuáles setrata de
constantes,generandoautomáticamentedireccionamiento directoo inmediato según corresponda. Anteuna
ambigüedad, existe la directiva OFFSET.En cuanto alos espaciosde memoria de datos, dada laarquitecturade
Holtek, limitaremosMCS-51 al manejo de memoriainterna,quegeneralmentees más rápidoy requiere instrucciones
más cortas.

Interrupción

Evaluamos el tiempo necesario enatenderuna interrupción, salvar contexto, y retomar ejecuciónnormal del
programa.En Holtek no disponemosde instruccionesparael stack, pero las instrucciones demove no afectanlos
flags.
Con uso del acumulador y flags:

push psw
push A
...
pop A
pop psw
reti

10 ciclos + overhead
Sin uso del acumulador ni flags (mover un pin en un port).

clr P2.1
setb P2.0
...
reti

4 ciclos + overhead

Con uso del acumulador y flags:

mov somewhere,A ; salva A
mov A,STATUS

mov somewhere2,A ; salva STATUS
...
mov A,somewhere2
mov STATUS,A
mov A,somewhere
reti

8 ciclos + overhead = 10 ciclos
Sin uso del acumulador ni flags (mover un pin en un port).

clr PA.1
set PA.0
...
reti

4 ciclos + overhead = 6 ciclos

Bifurcación (branch) según flag

Este es el típicocasodemodificacióndel flujo deprogramadebidoa circunstanciasespecificadasenun flag. Dada la
carenciade direccionamiento relativo,Holtek lo resuelvemedianteuna instrucción de salteo (skip)y un salto.
Generalmente, la bifurcación debe pensarse al revés

jnb bit,l1 ; si el bit está dentro del espacio de memoria direccionable como bits
...
jnb sfr.bit,l1 ; si se trata de un SFR direccionable por bits (ACC, B, ports, etc)
...
; caso contrario, debe llevarse a un SFR direccionable por bits (MOV, MOVX)

l1:

2 ciclos, 3 bytes

snz var.bit ; saltea instrucción si el bit 'bit' de la variable 'var' es 1
jmp l1 ; caso contrario salta a 'l1'
...

l1:

2/3 ciclos, 2 words

Salteo (skip) según flag

Esta situación seproducegeneralmentecuandodebemosalterarel estadodeun pin según nosindicaun flag, pero no
podemosdarnosel lujo de producir glitches (modificarloprimeroasumiendoun valor y despuéscorregirlo si el flag
indica otra cosa).Consideramos solamente el tiempoempleadoen la decisión.En MCS-51 asumimos que el flag está
en un SFR o posición de memoria direccionable por bits

jnb var.bit,l0
; acción si 1
sjmp l1

l0: ; acción si 0
l1:

4 ciclos, 5 bytes

sz var.bit
; acción si 1
snz var.bit
; acción si 0

3 ciclos, 2 words

  En un caso más simple, cuando sí nos podemos permitir el lujo de un glitch o simplemente modificamos un flag:
; acción si 0
jnb var.bit,l1
; acción si 1

l1:

2 ciclos, 3 bytes

; acción si 0
sz var.bit
; acción si 1

2 ciclos, 1 word

Modificar flag

  Alteramos el estado (seteamos o reseteamos) de un bit en una variable

setb var.bit ; si el flag está en el espacio de memoria/SFR direccionable como bits

orl var,(1<<bit) ; caso contrario. (1<<bit) puede no ser aceptado por el assembler

1(2) ciclos, 2(3) bytes

set var.bit

1 ciclo, 1 word

Típico loop

Consideramos solamente el overhead que introduce lo necesarioparaproducir el loop, es decir, decremento del
contador y salto al inicio del loop.

l1:
...
djnz R2,l1 ; decrementa R2 y si éste no es 0 vuelve al loop
; o bien
djnz var,l1 ; decrementa 'var' y si ésta no es 0 vuelve al loop

2 ciclos, 2 (3) bytes

l1:
...
sdz var ; decrementa 'var' y si ésta es 0, saltea la siguiente instrucción
jmp l1 ; vuelve al loop

3 ciclos, 2 words

Sumar valor a variable 8 bits

Sumamosun determinadovalor constante auna variablecualquiera;el resultadodebequedar en la variable en
cuestión.

mov A,#value
add A,var
mov var,A ; resultado de la suma en var

3 ciclos, 6 bytes

mov A,value
addm A,var ; resultado de la suma en var

2 ciclos, 2 words

  Lo mismo, pero el resultado queda en el acumulador 

mov A,#value
add A,var

2 ciclos, 4 bytes

mov A,value
add A,var ; resultado de la suma en A

2 ciclos, 2 words

Sumar una variable a otra, 8 bits

  Sumamos dos variables entre sí; el resultado debe quedar en la última variable direccionada.

mov A,var1
add A,var2
mov var2,A ; resultado de la suma en var2

3 ciclos, 6 bytes

mov A,var1
addm A,var2 ; resultado de la suma en var2

2 ciclos, 2 words

Memory move, inmediato 8 bits

  Colocamos un valor constante en una posición de memoria.

mov var,#value

1 ciclo, 2 bytes

mov A,value
mov var,A

2 ciclos, 2 words

Memory move, directo 8 bits

  Colocamos en una variable el valor contenido en otra.

mov var2,var1

2 ciclos, 3 bytes

mov A,var1
mov var2,A

2 ciclos, 2 words

Cálculo en 16 bits: K = next_T - advance – COMP_K

Realizamosun pequeño cálculono muy complejo,en el cualobtenemosel valor deuna variable deprocesoque
depende de otrasdosvariablesy una constante de calibración. Las variables son todas de 16-bits,y la constante de
calibración es de 8 bits, para mostrar un cálculo combinado.

clr C
mov A,next_Tl
subb A,advancel ; next_T – advance (LSB)
mov Kl,A
mov A,next_Th
subb A,advanceh ; MSB
mov Kh,A
mov A,Kl
clr C
subb A,COMP_K ; - COMP_K (8 bits)
mov Kl,A
jnc done ; carry ?
dec Kh ; sí, Kh = Kh – 1

done:

14 ciclos, 24 bytes

mov A,next_Tl
sub A,advancel ; next_T – advance (LSB)
mov Kl,A
mov A,next_Th
sbc A,advanceh ; MSB
mov Kh,A
mov A,Kl
sub A,COMP_K ; - COMP_K (8 bits)
mov Kl,A
snz C ; borrow ?
dec Kh ; sí, Kh = Kh – 1

done:

11 ciclos, 11 words

Cálculo simple en 32 bits: sumar dos variables

Sumamos dos variables en  32-bits, el resultado lo guardamos en una de ellas.

mov A,var1ll
add A,var2ll ; suma LSB
mov var2ll,A
mov A,var1lh
addc A,var2lh ; acarrea
mov var2lh,A
mov A,var1hl
addc A,var2hl
mov var2hl,A
mov A,var1hh
addc A,var2hh ; MSB
mov var2hh,A

12 ciclos, 24 bytes

mov A,var1ll
addm A,var2ll ; suma LSB
mov a,var1lh
adcm A,var2lh ; acarrea

mov a,var1hl
adcm A,var2hl
mov a,var1hh
adcm A,var2hh ; MSB

8 ciclos, 8 words

Máquina de estados

Analizamosa continuaciónun dispatcherdel tipo comúnmenteutilizado en máquinasdeestados,en el cual elflujo
de programaesredirigido según el valor deunavariable, es decir, elestadode lamáquina.En el casode Holtek, la
funcióny la tabla de saltos deberán estar dentro deun mismo segmento de 256words.En el casode MCS-51, existen
128 estados posibles como máximo.

mov DPTR,#STATETABLE
mov A,STATE ; estados posibles: 0,1,2,3,4,etc; 
add A,ACC ; x2, AJMP es una instrucción de 2 bytes
jmp @A+DPTR ; resultado de A+DPTR en PC

STATETABLE:
ajmp State1
ajmp State2

8 ciclos, 2 bytes por estado

mov A,STATE ; estados posibles: 0,1,2,3,4,etc
addm A,PCL ; resultado de la suma en PCL
jmp State1
jmp State2

5 ciclos, 1 word por estado

Table read, tabla en RAM

Tomamosun valor deunatabla según nosindica unavariable queoficia de índice dentro de lamisma.La tabla se
encuentraenRAM. En Holtek el tamañomáximo de la tabla es de 224bytes,en MCS-51,utilizandomemoriainterna,
dejando espacio para el stack y algunos registros, tal vez podamos extendernos algunos bytes más.

mov A,#TABLA ; tabla
add A,OFFSET ; + offset
mov R0,A
mov A,@R0 ; en memoria interna (el destino puede ser cualquier variable o SFR)

4 ciclos, 8 bytes, resultado en A

mov A,OFFSET TABLA ; tabla
add A,OFFST ; + offset
mov MP,A ; indirecto
mov A,IAR

4 ciclos, 4 words, resultado en A

Table read, tabla en memoria de programa

Tomamosun valor deunatabla según nosindica unavariable queoficia de índice dentro de lamisma.La tabla se
encuentraen ROM. En el casode Holtek, la tablapuedeocuparun espaciodentro del segmento de 256bytesen
memoria deprogramaen el que se estáejecutandoel código,o en el último delmapadememoria,las tablas nodeben
pasar un segmento de 256 unidades de memoria de programa.

mov DPTR,#TABLA
mov A,OFFSET
movc A,@A+DPTR
...

TABLA:

5 ciclos, 6 bytes+tabla, resultado en A, si se desea en var debe agregarse un mov adicional

mov A,OFFST
add A,TABLA
mov TBLP,A
tabrdl var ; tabla en última página de 256 bytes en memoria de programa
...

tablas .section at LASTPAGE 'code'
TABLA dc dato1, dato2,...

5 ciclos, 4 words+tabla, resultado en var (si se desea en A, var=ACC)

Direccionamiento indirecto en RAM

  Tomamos un valor de un buffer en RAM e incrementamos el puntero

mov R0,#TABLA ; tabla en memoria interna
mov A,@R0 ; el destino puede ser cualquier variable o SFR
inc R0

3 ciclos, 4 bytes

mov A,tabla
mov MP,A
mov A,IAR
inc MP ; incrementa puntero

4 ciclos, 4 words

Conversión binario a BCD

Finalmente,evaluaremosunasubrutinamuy común,comola empleadaparapresentarresultados enun displayde 7-
segmentos o LCD alfanumérico, es decir, la tradicional conversión de un byte a BCD.

BINBCD: mov R2,#8 ; COUNTER=8
mov RESULT,#0 ; RESULT=0

L1: mov A,SOURCE
rlc A
mov SOURCE,A ; rota SOURCE a la izquierda (MSB a Cy)
mov A,RESULT ; no afecta Cy
adc A,RESULT ; suma A + RESULT + Cy, resultado en A (A=2 x RESULT + Cy)
da A ; ajusta A para que sea BCD válido 
mov RESULT,A ; resultado en RESULT
djnz R2,L1 ; decrementa R2 y si éste no es 0 vuelve al loop
ret ; resultado en RESULT

76 ciclos, 19 bytes

BINBCD: mov A,8
mov COUNTER,A ; COUNTER=8
clr RESULT ; RESULT=0

L1: rlc SOURCE ; rota SOURCE a la izquierda (MSB a Cy)
mov A,RESULT ; no afecta Cy
adc A,RESULT ; suma A + RESULT + Cy, resultado en A (A=2 x RESULT + Cy)
daa RESULT ; ajusta A para que sea BCD válido, resultado en RESULT
sdz COUNTER ; decrementa 'COUNTER' y si éste es 0, saltea la siguiente instrucción
jmp L1 ; vuelve al loop
ret ; resultado en RESULT

60 ciclos, 10 words

